# AUTOMATED TDR TESTING MADE EASY WITH THE 7854 OSCILLOSCOPE/ 7S12 SAMPLER PLUG-IN



Tektronix®
COMMITTED TO EXCELLENCE

## Introduction

The principles of Time Domain Reflectometry (TDR) have long been well defined in theory. Simply send a pulse or step down a transmission line and observe the reflected signature of the line. It's something like radar in that the arrival time of the reflection indicates distance and the shape of the reflection indicates something about what caused the reflection. (See Appendix A)

Thanks to nanosecond – even subnanosecond – risetime systems, TDR techniques have become commonly accepted measurement practices.

Wherever transmission lines must be maintained, you're almost sure to find a Time Domain Reflectometer nearby. TDR today offers clean, clear signatures of cable faults with time resolutions that can pinpoint fault locations within fractions of an inch.

With capabilities such as this, you might think TDR had reached maturity as a measurement technique . . . but not yet! There's a new level of growth offered by digital signal processing. One of the basic techniques offered by digital signal processing is TDR difference testing, a technique that allows resistive component evaluation with accuracies approaching 0.0005 in reflection coefficient. For example, at a sensitivity setting of 50 mρ/div, reflection coefficient can be resolved to within 0.5 mρ.




**Figure 1:** TDR test setup using a feed-through sampler

## The Measurement System

An advanced TDR setup for digital signal processing is shown in fig.1.

It combines a Tektronix 7S12 TDR/Sampler plug-in with an S-52 Pulse Generator Head and an S-6 Sampling Head in a 7854 Waveform Processing Oscilloscope mainframe to form an ideal system for advanced TDR measurements. Details for connecting the 7S12 TDR/Sampler as a "loop thru" sampler are given in Fig. 2.

The combination shown in Fig. 2 provides a number of advantages to the user seeking accurate and reliable TDR measurement capabilities. The pulse source associated with the TDR/Sampler is tailored to TDR testing needs and is faster than most general-purpose pulse sources. The sampling techniques used in the 7S12 TDR/Sampler extend the useful range of the 7854 to 20 picoseconds per horizontal crt division. Even more importantly, the 7S12 TDR/Sampler is specific to TDR testing. When combined with the state-of-the-art digital storage and waveform processing capabilities of the 7854, it becomes an extremely powerful time domain reflectometer. (Fig.2.)




**Figure 2:** Test setup using the 7S12 TDR/Sampler in the 'loop thru' mode.

## Making Impedance Measurements with the 7854/7S12

The following basic relationships are used in making TDR impedance measurements:

$P = Er/Ei$

and

$Z_l = Z_o{*}(1 + \rho)/(1 - P)$

where $Z_l$ = unknown impedance
$Z_o$ = system impedance
$\rho$(RHO) = reflection coefficient
Er = reflected pulse
Ei = incident pulse

$Z_l$, the impedance along the line, is the unknown to be evaluated. The characteristic impedance of the system, $Z_o$, is accurately known and in the case of the 7S12 the impedance is 50 ohms. The 7S12 with precision matching pads can accomodate 75 and 93 ohm loads as well.

### Obtaining the waveform

Using the setup shown in Fig. 1, obtain the display of the incident waveform. The display should be similar to the waveform shown in Fig. 3. The settings of the m$\rho$/div on the 7S12 will vary according to the amount of deviation in the impedance of the device under test. Using the fine (zero set) control on the 7S12, the incident pulse can be set to the zero graticule line.

The waveform can now be digitized and stored by the 7854. The stored waveform consists of an array of 1024 waveform values. (The array size is programmable in the 7854.) Each value corresponds to the waveform amplitude at one of the 1024 equally spaced time increments across the display.

> **WARNING**
>
> **Coaxial cables stored on large reels may contain high-voltage (capacitive) charges resulting from manufacturer's preshipment tests or purchaser's acceptance tests. Other transmission lines may be charged with either dc or ac as a result of some line fault. It is important that TDR equipment operators take precautions to ensure that the test line has safely been discharged before it is handled or TDR equipment is attached to it.**

If everything were ideal, this waveform could be used to determine the impedance of the device under test (DUT). But real-life waveforms are often less than ideal. They are usually accompanied by additive noise and may be distorted by minor aberrations. These factors can contribute potential measurement errors that the user can minimize by using the features of the 7854 Waveform Processing Oscilloscope in the computation of the impedance of the DUT.

### Reducing Noise

To reduce the effects of random noise (Fig. 3), signal averaging can be used. Signal averaging is essentially the same as taking sequential acquisitions of the waveform, adding these acquisitions together and dividing by the number of acquisitions. To average 64 times with the 7854, simply enter '64 AVG' on the 7854 Waveform Calculator. The averaged waveform is observed on the 7854's STORED display. (See Fig. 4.) Averaging will improve the stored waveform's signal-to-noise ratio. In the case of random noise there is a 3 dB improvement in signal-to-noise ratio for each power of two averages. Thus, for 64 averages ($2^6$ averages) the improvement is $6 \times 3$ dB $= 18$ dB. For further improvement, more averages can be taken. However, the acquisition times will become correspondingly longer.

### Computing the Impedance by Simple Keystroke Entry.

Now that the waveform has been acquired (Fig. 4.) the actual impedance can be calculated. Turn on the two cursors on the 7854 by pressing the 'CRS2-1' key on the waveform calculator.




After positioning the cursors such that cursor 1 is on the portion of the waveform representing the 50 ohm line and cursor 2 is on a section of the waveform representing the DUT, enter the following key strokes.

| | |
|---|---|
| 1 enter | brings 1 into the X register. |
| VCRD + | This brings the vertical difference in m$\rho$ into the X register and adds it to 1. |
| 1 enter | brings 1 into the X register again and pushes the previous results into the Y register. |
| VCRD – | brings the vertical difference into the X register again and subtracts it from 1. The first result is in the Y register and the second result is in the X register. |
| / | divids the Y register by the X register. |
| 50 * | multiplies the value in the Y register by 50, the system impedance. |

The X register now contains the impedance of the point represented by cursor 2 in ohms (Fig. 5).

### Automated Testing using the 4052A with the 7854/7S12

An explanation of the tests in the program flow chart in appendix B follows. The three programs are impedance measurements using two cursors, distance measurements using two cursors and calculating average impedance. These measurements can be made either directly on the device under test (DUT), or by subtracting the waveform captured from a reference device from the DUT waveform. An example 7854 program is found in appendix D and the 4052A companion program is in appendix C.

### Impedance Measurement Using Two Cursors

This technique is useful in locating and measuring discrepancies in the DUT. The 7854/7S12 combination is easlily set up to capture the waveform for an impedance measurement. In the SCOPE mode, adjust the time-distance on the 7S12 to view the entire waveform, from the start of the 50 ohm reference line (015-1023-00) to the end of the DUT waveform (see Fig. 3). Adjust the mρ/div switch on the 7S12 so that the screen is about ¾ filled vertically. The algorithm used and an example of using it are found in the section 'Making Impedance Measurements' at the top of page 3. The following is a 7854 program that makes the impedance measurement. Enter it now.

000 64 AVG
001 STORED
002 CRS2-1
003 STOP
004 1 ENTER VCRD +
005 1 ENTER VCRD −/
006 50 *

Program remarks:

000 acquires a waveform by averaging it 64 times.
001 displays the averaged waveform (Fig. 4).
002 turns on two cursors.
003 halts the program. At this time the user adjusts the cursors on the stored waveform, cursor 1 to the 50 ohm reference and cursor 2 to a point to be measured.
004 1 is entered into the X register VCRD is the vertical difference between the cursors measured in mρ; + add 1 to this difference and the result is then pushed to the Y register by the next step.
005 1 is again entered into the X register, VCRD – subtracts VCRD from 1 then divides the Y register (1 VCRD +) by the X register (1 VCRD −).
006 multiplies the results of the previous operation by 50 (the reference impedance) (Fig. 5).




**Figure 3:** SCOPE MODE Noisy TDR Signal




**Figure 4:** STORED MODE 64 averages reduces the effects of random noise.

### Running the Program

To run this program on the 7854, enter 'f start' on the 7854 waveform calculator.




The 7854 readout will indicate that the program halted in line 3. Adjust the cursors and press 'run'. The result in ohms is displayed in the X register.




**Figure 5:** Impedance in ohms. Impedance measurement showing a value of 50.8Ω at cursor 2.




**Figure 6:** Distance in inches. Distance from cursor 1 to cursor 2 is 14.22 inches.

## Distance Measurements

To find the distance from a point on a waveform to a discontinuity, the algorithm is:

d = t*½V, where
- t = the time as measured from cursor 1 to cursor 2
- V = the propagation Velocity (C/E) where
  - C = 1. 18 × $10^{10}$ inches per sec.(speed of light)
  - E = the square root of the dielectric constant.

For Poly cable ½ V = 3.98 × $10^9$.

The setup and running of this program is the same as for the impedance program (Fig. 3). The user can easily identify the location of a connector to place the first cursor on, since connectors usually do show a larger discontinuity.

## Distance Measurement Program:

000 64 AVG
001 STORED
002 CRS2-1
003 STOP
004 HCRD
005 3.98 EEX9 *

000 acquires a signal by averaging it 64 times. (Fig. 4).
001 displays the acquired waveform.
002 turns on two cursors.
003 halts the program. The user now adjusts cursor 1 on a reference point and cursor 2 to the point to be measured.
004 brings the time between cursor 1 and 2 into the X register.
005 multiplies ½ the velocity of poly cable times the time. The result is in the X register in inches. (Fig. 6).

## Average Impedance

The user may want to know the average impedance of a DUT. Using the setup from the previous impedance measurement, the mean value of the 50 ohm line can be found and subtracted from the mean value of the DUT. This value is in mρ and can be used in the $Z_i = Z_o(1+\rho)/(1-\rho)$ formula. The 7854 program is as follows:

000 64 AVG
001 STORED
002 0 WFM 4 >WFM
003 CRS2 – 1
004 STOP
005 HXPD
006 OFF
007 MEAN 4 >CNS
008 4 WFM
009 CRS2 – 1
010 STOP

011 HXPD
012 OFF
013 MEAN 2 >CNS
014 4 CNS 2 CNS –
015 ABS 6 >CNS
016 1 ENTER 6 CNS +
017 1 ENTER 6 CNS –
018 /
019 50 *

**Program Remarks:**

000 Average 64 times.
001 Display averaged waveform.
002 Store averaged waveform in waveform location 4 (Fig. 7).
003 Turn on cursors.
004 Halts program. User now positions cursor 1 at the beginning of the 50 ohm section and cursor 2 at the end of the 50 ohm section of the waveform.
005 Horizontally expands the waveform between the two cursors.
006 Turns off the cursors.
007 Finds the mean of the waveform (total value of the points divided by the number of points), and stores value into 4 constant register (4 CNS). (Fig. 8).
008 Recall original WFM.
009 Turn on cursors.
010 Halts program. User now positions cursor 1 at the beginning of the DUT waveform and cursor 2 at the end of the DUT waveform.
011 Horizontally expands the waveform between the two cursors.
012 Turn off the cursors.
013 Find the mean of this waveform, put into 2 constant register. (Fig. 9).
014 Subtract 2 CNS from 4 CNS.
015 Take the absolute value and store it in 6 CNS.
016 Add 1 to 6 CNS.
017 Subtract 6 CNS from 1 (1 + is in the Y register 1 – is in the X register).
018 Divide the Y register by the X register.
019 Multiply this value by 50 ohms and the X register contains the average impedance (Fig. 10).

## Using a Reference Waveform

The user may find that when storing the waveforms of a DUT, there will be system discontinuities (caused by connectors, cabling, etc.) that are of no interest in the measurement. One way to remove those discontinuities is to store away the waveform captured from a known good device, remove that device and capture a waveform from a device to be tested, and then remove common discontinuities by waveform subtraction.




**Figure 7:** Waveform averages 64 times.




**Figure 8:** Mean of horizontally expanded 50 ohm secion of waveform.

The following program is an example of this technique. In using this program as shown in the flow chart in appendix B, the menu would be used to jump into the impedance program or distance program at a point after the waveform acquisition.

```
000  64 AVG STORED
001  0 WFM 3 >WFM
002  STOP
003  64 AVG
004  0 WFM 3 WFM –
005  0WFM 2 >WFM
```

000 Average the waveform 64 times and display.
001 Store the averaged waveform in waveform location 3 .
002 Halt program, remove the reference waveform and connect the DUT.
003 Average the waveform 64 times.
004 Subtract the previous waveform (3 WFM) from the DUT waveform.
005 Store the results in waveform location 2.



**Figure 9:** Mean of horizontally expanded section of DUT.



**Figure 10:** Averaged impedance in ohms.



**Figure 11:** 7S12 Scope Mode Display

### Controller vs. stand alone

The 7854/7S12 will perform this testing with or without a controller, however a controller does give the user many advantages. A controller can provide the ability to transfer instructions to the 7854 crt, thereby prompting the user on the next test sequence. The 7854 program can also be stored and recalled by the controller, for example the 4052A mag tape could store a 7854 program. By pressing a User Definable Key,(UDK), the program could be sent back to the 7854. A menu of tests that can be done on the 7854 could also be printed by the controller. In this case, the 4052A would print an entire list of tests and the user would simply press the appropriate UDK to select a test to be run. Appendixes C and D together make up such a program, with appendix B showing the interaction of the two programs.

Using the 7854/7S12 as a stand-alone unit to perform a series of tests would require an instruction sheet to tell the operator (a) how to start a specific test and (b) what to do at each step of the test. The following is an example of what this instruction sheet would be like for calculating the average impedance (using the same 7854 program in appendix D):

(c) average impedance
- a. set up the 7S12 (see Fig. 11)
- b. press '3 LBL GOTO' then 'RUN'
- c. move the cursors to bracket the 50 ohm section of the waveform
- d. press 'RUN'
- e. move the cursors to bracket the DUT section of the waveform
- f. press 'RUN'
- g. the average impedance is printed in the X register

## Loading the 7854 Program

Any 7854 program can be keyed into the 7854 from the Waveform Calculator keyboard. In the automated test setup, it may not be feasible to enter the program in appendix B from the keyboard every morning at instrument power up. After ensuring that the 7854 program works properly, the program can be stored on tape for recall.

## Summary

Automated TDR testing of devices using a 4052A, 7854 and 7S12 has many advantages. Waveform averaging reduces the signal-to-noise ratio. Operator instructions can be displayed on the 7854 crt to help prevent errors or confusion. Complex 7854 programs can be stored and recalled from the 4052A mag tape. A series of tests can be done on a device or tests can be repeated. There is no end to the variety of uses, including logging data to the mag tape or having either the 7854 or 4052A make pass/fail decisions on the results from a test.

## Appendix A

## TDR Basics

TDR, short for Time Domain Reflectometry, is a pulse testing technique for evaluating wide-band transmission systems. The central idea is to send a pulse through a transmission system, which commonly consists of simply a transmission line and a termination but may also include other devices. The transmitted pulse, referred to as the incident pulse, travels down the line. Then, according to the characteristics of the line and its termination, portions of the pulse are absorbed or reflected. By studying and measuring the reflections (reflectometry), certain characteristics of the system can be determined.




A simple TDR setup.




A simple system illustrating TDR principles is shown in the accompanying diagram. It consists of a pulse generator with a 50 ohm output, a high-impedance oscillo scope (7854), a ten or twenty-foot length of coaxial cable with a 50 ohm characteristic impedance (RG-58C/U for example), a terminating resistance and the appropriate connectors for hooking up the system.

Once you have the system hooked up, remove the terminating resistor ($R_1$ becomes ϰ). Then adjust the oscilloscope for a display similar to the one shown in the following crt photo.




Reflection of a transmission line terminated in an open.

The first transition in the open-circuit display is the incident step. The second transition is the reflected step, and it occurs at a point corresponding to the end of the cable. The time between the incident and reflected steps is twice the propagation time for a one-way trip through the cable. This time, $t_r$, represents the physical length of the cable, which can be computed by the following formula.

$$\text{length} = \frac{C\,t_r}{2\sqrt{\varepsilon}}$$

In this formula, C is the velocity of light and ε, is the effective dielectric constant of the dielectric material in the transmission line.

For the open-circuit example, nearly all of the incident energy is reflected — there is no termination to absorb the energy of the incident step. In the case of an ideal cable, there are no losses, and an open circuit results in total reflection. However, real-life cables do have losses, and there is some attenuation of the transmitted signal. Thus, the reflected step in the crt photo doesn't quite equal the incident step.

The second crt photo is a composite of reflections for various terminations. These terminations are labeled on the photo along with their corresponding reflection coefficients.

The voltage reflections for various resistive terminations are generally expressed in terms of a reflection coefficient, where

$$\rho = {}^{H}\text{reflected}/{}^{H}\text{incident}$$

Also the reflection coefficient may be expressed in terms of the line's characteristic impedance, $Z_o$, and termination, $R_l$.

$$\rho = (R_l - Z_o)/R_l + Z_o)$$

Thus, it is possible to convert TDR measurements into absolute measurements of $R_l$ if $Z_o$ is known precisely. Or, if a standard $R_l$ is used to obtain a calibration reflection, an unknown $R_l$ can be evaluated by compairing its reflection with the standard. The latter technique is highly accurate when used with the digital storage and waveform processing capabilities of the 7854 Oscilloscope.

## APPENDIX B
### 4052A/7854 FLOW CHART



## APPENDIX C

**4052A Program:**

```
1 GO TO 100
4 WBYTE @A+32,20:
5 WBYTE @A+32,1,63:
7 GO TO 490
8 WBYTE @A+32,20:
11 GO TO 1060
12 WBYTE @A+32,20:
15 GO TO 1470
16 WBYTE @A+32,20:
19 GO TO 1910
20 WBYTE @A+32,20:
21 GO TO 2410
36 WBYTE @A+32,20:
37 WBYTE @A+32,1,17:
39 END
40 WBYTE @A+32,20:
43 GO TO 210
100 REM INPUT THE 7854
110 REM address
120 PAGE
130 PRINT "J J INPUT THE DEVICE";
140 PRINT "number of the 7854I_";
150 INPUT A
160 REM INITIALIZE THE 7854
170 WBYTE @A+32,20:
180 PRINT @A:"RQSON IOCON REMON"
190 PRINT @A:"exron ceron opcon"
200 REM MENU
210 PAGE
220 PRINT "J J I *********************"
230 PRINT "I           MENU"
240 PRINT "I*********************"
250 PRINT "J PRESS THE USER DEFINABLE";
260 PRINT "KEY (UDK) FOR THE"
270 PRINT "    APPROPRIATE SELECTION"
280 PRINT "---------------------------------------"
290 PRINT "program must be loaded first (UDK #5)"
300 PRINT "---------------------------------------"
310 PRINT "J #1...CAPTURE A REFERENCE WAVEFORM"
320 PRINT " to use with other tests"
330 PRINT "J #2...MEASURE IMPEDANCE"
340 PRINT " point to point using cursors"
350 PRINT "J #3...MEASURE DISTANCE"
360 PRINT " point to point using cursors"
370 PRINT "J #4...MEASURE AVERAGE IMPEDANCE"
380 PRINT " using cursors to bracket";
390 PRINT " areas to be measured"
400 PRINT "J #5...LOAD 7854 PROGRAM"
410 PRINT "J #9...SET 7854 TO LOCAL (GTL)"
420 PRINT "J #10..RE-PRINT MENU"
430 S=0
440 DIM T$(1000)
450 R$=CHR(13)
460 WBYTE @A+32,20:
470 WBYTE @A+32,1,63:
480 END
490 REM CAPTURING A REFERANCE WAVEFORM
500 PAGE
510 ON SRQ THEN 2640
520 PRINT @A:"1 LBL GOTO RUN"
530 IF S<>66 THEN 530
540 S=0
550 T$="Connect a reference DUT to the"
560 T$=T$&R$
570 T$=T$&"7S12. The 7S12 should have already"
580 T$=T$&R$
590 T$=T$&"have been setup according to the "
600 T$=T$&R$
610 T$=T$&"manual. Set the mp/div, dc offset"
620 T$=T$&R$
630 T$=T$&"and time/div of a resonable display"
640 T$=T$&R$
650 T$=T$&"Press 'RQS' on 7854 to continue"
660 GOSUB 2560
670 WBYTE @A+32,1,63:
680 IF S<>80 THEN 680
690 PRINT @A:"RUN"
700 IF S<>66 THEN 700
710 T$="Now disconnect the reference"
720 T$=T$&R$
730 T$=T$&"signal and connect the DUT"
740 T$=T$&R$
750 T$=T$&"Press 'RQS' on the 7854 to continue"
760 GOSUB 2560
770 IF S<>80 THEN 770
780 T$=""
790 GOSUB 2560
800 PRINT @A:"RUN"
810 IF S<>66 THEN 810
820 S=0
830 PAGE
840 PRINT "J J INPUT THE TEST TO BE";
850 PRINT "DONE NEXT"
860 PRINT "J I input  1 for point to point impedance'
870 PRINT "J I input  2 for point";
880 PRINT "to point impedance"
890 PRINT "J I input  3 for average";
900 PRINT "impedance"
910 WBYTE @A+32,1,63:
920 INPUT T
930 IF T=3 THEN 1980
940 T$="1"&STR$(T)
950 PRINT @A:T$;" LBL GOTO RUN"
960 IF S<>66 THEN 960
970 S=0
980 WBYTE @A+32,1,63:
990 PRINT "J J Do you wish to make";
1000 PRINT "further measurements on this"
1010 PRINT "waveform? If you do then input";
1020 PRINT "'y', if not then 'n'"
1030 INPUT X$
1040 IF X$="y" THEN 830
1050 END
1060 REM MAKING POINT TO POINT
1070 REM IMPEDANCE MEASUREMENTS
1080 PAGE
1090 ON SRQ THEN 2640
1100 PRINT @A:"2 LBL GOTO RUN"
1110 IF S<>66 THEN 1110
1120 S=0
1130 T$="Set the 7S12 controls"
1140 T$=T$&"for appropriate"
1150 T$=T$&R$
1160 T$=T$&"display, press 'RQS'"
1170 T$=T$&"on the 7854"
1180 T$=T$&R$
1190 T$=T$&"to continue"
1200 GOSUB 2560
1210 WBYTE @A+32,1,63:
1220 IF S<>80 THEN 1220
1230 PRINT @A:"RUN"
1240 IF S<>66 THEN 1240
1250 S=0
1260 T$="Now set cursor 1 to"
1270 T$=T$&"the baseline"
1280 T$=T$&R$
1290 T$=T$&"and cursor 2 to"
1300 T$=T$&"the point to be"
1310 T$=T$&R$
1320 T$=T$&"measured. press 'RQS'"
1330 T$=T$&"on the 7854"
1340 T$=T$&R$
```

## APPENDIX C (continued)

```
1350 T$=T$&"to continue."
1360 GOSUB 2560
1370 WBYTE @A+32,1,63:
1380 IF S<>80 THEN 1380
1390 PRINT @A:"run"
1400 IF S<>66 THEN 1400
1410 S=0
1420 PRINT @A:"sendx"
1430 IF S<>210 THEN 1430
1440 INPUT @A:X
1450 PRINT "Impedance is ";X
1460 END
1470 REM MAKING POINT TO POINT
1480 REM DISTANCE MEASUREMENTS
1490 PAGE
1500 ON SRQ THEN 2640
1510 PRINT @A:"3 LBL GOTO RUN"
1520 IF S<>66 THEN 1520
1530 S=0
1540 DIM T$(1000)
1550 R$=CHR(13)
1560 T$="Set the 7S12 controls"
1570 T$=T$&"for appropriate"
1580 T$=T$&R$
1590 T$=T$&"display, press 'RQS'"
1600 T$=T$&"on the 7854"
1610 T$=T$&R$
1620 T$=T$&"to continue"
1630 GOSUB 2560
1640 WBYTE @A+32,1,63:
1650 IF S<>80 THEN 1650
1660 PRINT @A:"RUN"
1670 IF S<>66 THEN 1670
1680 S=0
1690 T$="Set cursor 1 to a"
1700 T$=T$&"reference point"
1710 T$=T$&R$
1720 T$=T$&"set cursor 2 to"
1730 T$=T$&"the point to be"
1740 T$=T$&R$
1750 T$=T$&"measured. Press 'RQS'"
1760 T$=T$&"on the 7854"
1770 T$=T$&R$
1780 T$=T$&"to continue."
1790 GOSUB 2560
1800 WBYTE @A+32,1,63:
1810 IF S<>80 THEN 1810
1820 PRINT @A:"RUN"
1830 IF S<>66 THEN 1830
1840 S=0
1850 PRINT @A:"SENDX"
1860 IF S<>210 THEN 1860
1870 S=0
1880 INPUT @A:X
1890 PRINT "measured distance is ";X
1900 END
1910 REM MAKING AVERAGE IMPEDANCE
1920 REM MEASUREMENTS
1930 PAGE
1940 ON SRQ THEN 2640
1950 PRINT @A:"4 LBL GOTO RUN"
1960 IF S<>66 THEN 1960
1970 S=0
1980 T$="Set the 7S12 controls"
1990 T$=T$&"for appropriate"
2000 T$=T$&R$
2010 T$=T$&"display. Press 'RQS'"
2020 T$=T$&"on the 7854 to"
2030 T$=T$&R$
2040 T$=T$&"continue."
2050 GOSUB 2560
2060 WBYTE @A+32,1,63:
2070 IF S<>80 THEN 2070
2080 PRINT @A:"RUN"
2090 IF S<>66 THEN 2090
2100 S=0
2110 T$="Set the cursors to"
2120 T$=T$&"bracket the baseline"
2130 T$=T$&R$
2140 T$=T$&"Press 'RQS' on the"
2150 T$=T$&"7854 to continue"
2160 GOSUB 2560
2170 WBYTE @A+32,1,63:
2180 IF S<>80 THEN 2180
2190 PRINT @A:"run"
2200 IF S<>66 THEN 2200
2210 S=0
2220 T$="Set the cursors to bracket"
2230 T$=T$&"the DUT waveform"
2240 T$=T$&R$
2250 T$=T$&"of interest. Press 'RQS'"
2260 T$=T$&"on the 7854 to"
2270 T$=T$&R$
2280 T$=T$&"continue."
2290 GOSUB 2560
2300 WBYTE @A+32,1,63:
2310 IF S<>80 THEN 2310
2320 PRINT @A:"RUN"
2330 IF S<>66 THEN 2330
2340 S=0
2350 PRINT @A:"SENDX"
2360 IF S<>210 THEN 2360
2370 S=0
2380 INPUT @A:X
2390 PRINT "Average impedance is ";X
2400 END
2410 REM TAPE LOADING ROUTINE
2420 ON SRQ THEN 2640
2430 FIND 2
2440 S=0
2450 I=0
2460 PRINT @A:"PROGRAM CLP NEXT"
2470 IF S<>0 THEN 2470
2480 ON EOF (0) THEN 2530
2490 INPUT @33:A$
2500 PRINT @A:A$
2510 I=I+1
2520 GO TO 2490
2530 PRINT @A:"EXECUTE"
2540 IF S<>66 THEN 2540
2550 END
2560 REM SHIP TEXT TO THE 7854
2570 S=0
2580 PRINT @A:"EXECUTE >TEXT"
2590 IF S<>149 AND S<>213 THEN 2590
2600 PRINT @A:T$
2610 IF S<>66 THEN 2610
2620 S=0
2630 RETURN
2640 REM POLL ROUTINE
2650 POLL D,S;A
2660 RETURN
```

## APPENDIX D
### 7854 Program:

### Subtracting the Reference Waveform from the DUT Waveform.

```
000 L01
001 VMDR HMDA OFF STORED STOP
002 STORED
003 64 AVG 0 WFM 2 >WFM STOP
004 STORED
005 64 AVG 0 WFM 3 >WFM
006 2 WFM 3 WFM - 0 >WFM
007 CRS2-1
008 STOP
```

### 7854 Program Impedance Measurements

```
009 L02
010 VMDR HMDA OFF STORED STOP
011 CRS2-1 STORED
012 64 AVG STOP
013 L11
014 VCRD 1 + VCRD 1 - /
015 ABS 50 *
016 STOP
```

### Distance Measurement (Poly Cable).

```
017 L03
018 VMDR HMDA OFF STORED STOP
019 STORED
020 64 AVG
021 CRS2-1 STOP
022 L12
023 HCRD 3.98EEX9 *
024 STOP
```

### Average Impedance Measurements.

```
025 L04
026 VMDR HMDA OFF STORED STOP
027 STORED
028 64 AVG
029 L13
030 CRS2-1 STOP
031 0 WFM 4 >WFM
032 HXPD OFF MEAN 4 >CNS
033 CRS2-1 4 WFM
034 STOP
035 HXPD OFF MEAN 2 >CNS
036 4 CNS 2 CNS -
037 IFY>X 44 LBL GOTO
038 4 CNS 2 CNS - CHS6 >CNS
039 45 LBL GOTO
040 L44
041 4 CNS 2 CNS - ABS 6 >CNS
042 L45
043 6 CNS 1 + 6 CNS 1 - /
044 ABS 50 *
045 STOP
```


**For further information, contact:**

**U.S.A., Asia, Australia, Central & South America, Japan**
Tektronix, Inc.
P.O. Box 1700
Beaverton, Oregon 97075
For additional literature, or the address and phone number of the Tektronix Sales Office nearest you, contact:
Phone: 800/547-1512
Oregon only 800/452-1877
TWX: 910-467-8708
TLX: 15-1754
Cable: TEKTRONIX

**Europe, Africa, Middle East**
Tektronix Europe B.V.
European Headquarters
Postbox 827
1180 AV Amstelveen
The Netherlands
Phone: (20) 471146
Telex: 18312 - 18328

**Canada**
Tektronix Canada Inc.
P.O. Box 6500
Barrie, Ontario L4M 4V3
Phone: 705/737-2700

**Tektronix sales and service offices around the world:**
Albania, Algeria, Angola, Argentina, Australia, Austria, Bangladesh, Belgium, Bolivia, Brazil, Canada, Peoples Republic of China, Chile, Colombia, Costa Rica, Czechoslovakia, Denmark, East Africa, Ecuador, Eqypt, Federal Republic of Germany, Finland, France, Greece, Hong Kong, Hungary, Iceland, India, Indonesia, Ireland, Israel, Italy, Ivory Coast, Japan, Jordan, Korea, Kuwait, Lebanon, Malaysia, Mexico, Morocco, The Netherlands, New Zealand, Nigeria, Norway, Pakistan, Panama, Peru, Philippines, Poland, Portugal, Qatar, Republic of South Africa, Romania, Saudi Arabia, Singapore, Spain, Sri Lanka, Sudan, Sweden, Switzerland, Syria, Taiwan, Thailand, Turkey, Tunisia, United Kingdom, Uruguay, USSR, Venezuela, Yugoslavia, Zambia, Zimbabwe.




**Tektronix**®
COMMITTED TO EXCELLENCE


5/83 42W-5334